NATIONAL AERONAUTICS AND SPACE ADMINISTRATION

Technical Report 32-1338

Near-Encounter Geometry Generation

T. C. Duxbury

JET PROPULSION LABORATORY
CALIFORNIA INSTITUTE OF TECHNOLOGY
PASADENA, CALIFORNIA

November 15, 1968

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION

Technical Report 32-1338

Near-Encounter Geometry Generation

T. C. Duxbury

JET PROPULSION LABORATORY
CALIFORNIA INSTITUTE OF TECHNOLOGY
PASADENA, CALIFORNIA

November 15, 1968

TECHNICAL REPORT 32-1338

Prepared Under Contract No. NAS 7-100
National Aeronautics & Space Administration

Preface

The work described in this report was performed by the Guidance and Control Division of the Jet Propulsion Laboratory.

Contents

Tables

Figures

Abstract

A method is presented for generating the near-encounter spacecraft–target-planet celestial geometry by including the motion of the spacecraft–target-planet system about the sun in the two-body trajectory solution. This method has been incorporated in a computer program that has been used to simulate an earth–Mars trajectory. The computer program, which uses output data from the SPACE–Single Precision Cowell Trajectory Program as input, is presented along with the results from the trajectory studied.

Near-Encounter Geometry Generation

I. Introduction

A major effort of mission analysis is concerned with the near-target-planet encounter phase. Knowledge of the pre- and post-encounter trajectory geometry is necessary to determine sensor look angles, science and antenna pointing requirements, occultation zones, and guidance constraints. This report describes an approximate method for generating the near-encounter geometry to an accuracy greater than 0.1 deg during an interval of up to 20 days from encounter. The approximate method, not requiring integration of the spacecraft equations of motion, can generate a complete trajectory geometry for a 20-day period in a matter of minutes when programmed on the IBM-1620 computer.

Near-encounter motion of a spacecraft with respect to a target planet will be described by the hyperbola defined in a planet-centered, inertial coordinate system (RST) at planet encounter. Data containing the inertial direction in **RST** to celestial bodies and the classical elements defining the hyperbola used in the analyses are available in the target planet-centered conic section of the SPACE–Single Precision Cowell Trajectory Program output (Ref. 1). The Appendix lists an IBM-1620 computer program that calculates the target-planet cone angle, clock angle, phase angle, and angular diameter; spacecraft–target-planet range; target-planet–spacecraft–Canopus angle; and Canopus cone angle.

II. Two-Body Trajectory Solution

Figure 1 shows a general planet-centered spacecraft vector **r*** in the **RST** coordinate system with

$$\mathbf{r} = r \begin{bmatrix} \cos\theta\cos i\sin\omega' + \sin\theta\cos\omega' \\ \sin i\sin\omega' \\ \cos\theta\cos\omega' - \sin\theta\cos i\sin\omega' \end{bmatrix} \tag{1}$$

where

R = vector completing right-hand system

S = vector along incoming asymptote

T = vector parallel to ecliptic plane

ω = argument of periapsis

$\omega' = \omega + \eta$ (η = true anomaly)

i = inclination of trajectory plane to **T-R** plane

θ = orientation of impact parameter **B** in **T-R** plane

r = target planet–spacecraft range

*Vectors are differentiated as follows: $\mathbf{r}$ = vector; $\hat{\mathbf{r}}$ = unit vector.



Fig. 1. Target-planet–spacecraft vector

For a specific trajectory with the accompanying **RST** coordinate system,

$$\begin{aligned} i &= 90 \text{ deg} \\ \omega &= \tan^{-1}(a/B) \end{aligned} \qquad (2)$$

where

a = semimajor axis of hyperbola

B = magnitude of **B**

These definitions of i and ω would not necessarily hold, for example, when assuming an **RST** coordinate system and then defining the actual trajectory in the assumed system as would be done in an orbit determination problem.

For a target planet with a gravitational constant μ and radius *rad*, and for a hyperbolic trajectory defined by the aim point **B** and hyperbolic excess velocity V_∞, the following equations hold for the two-body trajectory solution (Ref. 2).

$$a = \mu/V_\infty^2 \qquad (3)$$

$$e = \left(\frac{a^2 + B^2}{a^2}\right)^{1/2} = \text{hyperbolic eccentricity} \qquad (4)$$

$$\tau = \left(\frac{a^3}{\mu}\right)^{1/2} (e \sinh F - F) = \text{time past encounter} \qquad (5)$$

$$\begin{aligned} F &= \cosh^{-1}[(a + r)/ae] \operatorname{sgn}(\tau) \\ &= \text{hyperbolic eccentric anomaly} \end{aligned} \qquad (6)$$

$$\eta = \cos^{-1}[(B^2 - ra)/rae] \operatorname{sgn}(\tau) \qquad (7)$$

Given a time τ, Eq. (5) can be solved by an iterative process, such as the Newton–Raphson technique (Ref. 3), to yield r. Equation (1), defining the vector **r**, can then be solved. The angle subtended by the target planet as seen from the spacecraft (planet angular diameter) is given by

$$\Phi = 2 \sin^{-1}(rad/r) \qquad (8)$$

III. Sun Direction in RST

To define the spacecraft-centered celestial cone angle β and clock angle α of the target planet (Fig. 2), it is necessary to define the direction to the sun and Canopus in the target-planet **RST** system and then transform these directions to the spacecraft at time τ.



Fig. 2. Celestial coordinate system

Figure 3 defines the sun direction $\hat{\mathbf{R}}_p$ and Canopus direction $\hat{\mathbf{C}}$ in the target-planet **RST** system at encounter ($\tau = 0$).

At a time τ, other than zero, the inertial direction to the sun will have changed due to the target-planet's motion about the sun. This change in direction can be expressed as a rotation of $\mathbf{R}_p$ about a vector **W**, normal to the target-planet plane of motion as a function of τ.



Fig. 3. Sun and Canopus directions



Fig. 4. VUW coordinate system

Thus, the inertial **VUW** coordinate system, shown in Fig. 4, is defined such that $\mathbf{V} = \mathbf{R}_p$ at $\tau = 0$, and **U** completes the right-hand system.

Sun vector $\mathbf{R}_p$ defined in **VUW** at time τ is given by

$$\mathbf{R}_p = R_p(\tau)\begin{bmatrix} \cos\lambda(\tau) & -\sin\lambda(\tau) & 0 \\ \sin\lambda(\tau) & \cos\lambda(\tau) & 0 \\ 0 & 0 & 1 \end{bmatrix}\begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix}$$

$$= R_p(\tau)\,\mathbf{J}(\tau)\,\hat{\mathbf{V}} \tag{9}$$

where

$\lambda(\tau) = \int_0^\tau \dot{\lambda}(t)\,dt$

$\dot{\lambda}(\tau)$ = target-planet angular rate about the sun

$R_p(\tau)$ = sun–target-planet range

For the time period in which the two-body approximation holds

$$\dot{\lambda}(\tau) \approx C/R_p^2 \tag{10}$$

where

C = Kepler's area constant

and

$$R_p = R_p(0)$$

Therefore,

$$\lambda(\tau) \approx C\,\tau/R_p^2 \tag{11}$$

The transformation K from **VUW** to **RST** is needed to express $\mathbf{R}_p$ in **RST**. An intermediate coordinate system **XYZ** will be used in determining K. In **RST**,

$$\begin{bmatrix} \hat{\mathbf{X}} \\ \hat{\mathbf{Y}} \\ \hat{\mathbf{Z}} \end{bmatrix} = \begin{bmatrix} \hat{\mathbf{R}}_p(0) \\ \hat{\mathbf{S}} \\ \dfrac{\hat{\mathbf{R}}_p(0)\times\hat{\mathbf{S}}}{|\hat{\mathbf{R}}_p(0)\times\hat{\mathbf{S}}|} \end{bmatrix} = \begin{bmatrix} -(\sin\xi_s\sin\eta_s) & (\cos\xi_s) & -(\sin\xi_s\cos\eta_s) \\ 0 & 1 & 0 \\ \cos\eta_s & 0 & -\sin\eta_s \end{bmatrix}\begin{bmatrix} \hat{\mathbf{R}} \\ \hat{\mathbf{S}} \\ \hat{\mathbf{T}} \end{bmatrix} \tag{12}$$

or

$$\begin{bmatrix} \hat{\mathbf{X}} \\ \hat{\mathbf{Y}} \\ \hat{\mathbf{Z}} \end{bmatrix} = L \begin{bmatrix} \hat{\mathbf{R}} \\ \hat{\mathbf{S}} \\ \hat{\mathbf{T}} \end{bmatrix} \tag{13}$$

In **VUW**,

$$\begin{bmatrix} \hat{\mathbf{X}} \\ \hat{\mathbf{Y}} \\ \hat{\mathbf{Z}} \end{bmatrix} = \begin{bmatrix} \hat{\mathbf{R}}_p(0) \\ \hat{\mathbf{S}} \\ \dfrac{\hat{\mathbf{R}}_p(0) \times \hat{\mathbf{S}}}{|\hat{\mathbf{R}}_p(0) \times \hat{\mathbf{S}}|} \end{bmatrix} = \begin{bmatrix} 1 & 0 & 0 \\ (\cos\gamma_p \cos\sigma_p) & (\cos\gamma_p \sin\sigma_p) & (\sin\gamma_p) \\ 0 & -(\sin\gamma_p/\sin\xi_s) & (\cos\gamma_p \sin\sigma_p/\sin\xi_s) \end{bmatrix} \begin{bmatrix} \hat{\mathbf{V}} \\ \hat{\mathbf{U}} \\ \hat{\mathbf{W}} \end{bmatrix} \tag{14}$$

or

$$\begin{bmatrix} \hat{\mathbf{X}} \\ \hat{\mathbf{Y}} \\ \hat{\mathbf{Z}} \end{bmatrix} = M \begin{bmatrix} \hat{\mathbf{V}} \\ \hat{\mathbf{U}} \\ \hat{\mathbf{W}} \end{bmatrix} \tag{15}$$

The transformation K can be expressed in terms of L and M as follows:

$$K = L^{-1}M \tag{16}$$

where

$$K = \begin{bmatrix} -\sin\xi_s \sin\eta_s & \dfrac{\cos\xi_s \sin\eta_s \cos\gamma_p \sin\sigma_p - \cos\eta_s \sin\gamma_p}{\sin\xi_s} & \dfrac{\cos\xi_s \sin\eta_s \sin\gamma_p + \cos\eta_s \cos\gamma_p \sin\sigma_p}{\sin\xi_s} \\ \cos\xi_s & \cos\gamma_p \sin\sigma_p & \sin\gamma_p \\ -\sin\xi_s \cos\eta_s & \dfrac{\cos\xi_s \cos\eta_s \cos\gamma_p \sin\sigma_p + \sin\eta_s \sin\gamma_p}{\sin\xi_s} & \dfrac{\cos\xi_s \cos\eta_s \sin\gamma_p - \sin\eta_s \cos\gamma_p \sin\sigma_p}{\sin\xi_s} \end{bmatrix}$$

The **XYZ** coordinate system is not necessarily an orthogonal coordinate frame. Also, the angle σ_p is not included in the SPACE output. From the geometry, however,

$$\cos\sigma_p = \cos\xi_s/\cos\gamma_p \tag{17}$$

In general, outer planets have a greater angular velocity about the sun than the approaching spacecraft for type I trajectories (Ref. 4). The inverse would apply to trajectories to the inner planets. Under these conditions, the sign of $\sin\sigma_p$ may be determined by using $0 \leq \sigma_p < 180$ deg for type I trajectories to the outer planets, and $180 \leq \sigma_p < 360$ deg for type I trajectories to the inner planets.

The sun vector $\mathbf{R}_p$ at time τ and expressed in **RST** is then

$$\mathbf{R}_p = R_p K \mathbf{J}(\tau) \hat{\mathbf{V}} \tag{18}$$

Using Eq. (18), the phase angle of the target planet as seen from the spacecraft is given by

$$\phi = \cos^{-1}(\hat{\mathbf{R}}_p \cdot \hat{\mathbf{r}}) \tag{19}$$

IV. Spacecraft-Centered Celestial Geometry

The direction to Canopus $\hat{\mathbf{C}}$ can be considered constant and equal in either the planet-centered or spacecraft-centered **RST** system, because of the great distance to Canopus. The separation angle δ between the target planet and Canopus as viewed from the spacecraft is given by

$$\delta = \cos^{-1}(-\hat{\mathbf{r}} \cdot \hat{\mathbf{C}}) \tag{20}$$

where

$$\hat{\mathbf{C}} = -(\sin \xi_c \sin \eta_c)\,\mathbf{R} + (\cos \xi_c)\,\mathbf{S} - (\sin \xi_c \cos \eta_c)\,\mathbf{T}$$

The vector **c** in the spacecraft-centered celestial **abc** coordinate system (Fig. 2) is defined by transforming the sun direction to the spacecraft-centered **RST** system. Therefore, in spacecraft coordinates,

$$\hat{\mathbf{c}} = \frac{\mathbf{R}_p - \mathbf{r}}{|\mathbf{R}_p - \mathbf{r}|}$$

and by definition

$$\left.\begin{aligned} \hat{\mathbf{b}} &= \frac{\hat{\mathbf{c}} \times \hat{\mathbf{C}}}{|\hat{\mathbf{c}} \times \hat{\mathbf{C}}|} \\ \hat{\mathbf{a}} &= \hat{\mathbf{b}} \times \hat{\mathbf{c}} \end{aligned}\right\} \tag{21}$$

which can be expressed as follows:

$$\begin{bmatrix} \hat{\mathbf{a}} \\ \hat{\mathbf{b}} \\ \hat{\mathbf{c}} \end{bmatrix} = N \begin{bmatrix} \hat{\mathbf{R}} \\ \hat{\mathbf{S}} \\ \hat{\mathbf{T}} \end{bmatrix} \tag{22}$$

The cone angle of Canopus β_c is given by

$$\beta_c = \cos^{-1}(\hat{\mathbf{c}} \cdot \hat{\mathbf{C}}) \tag{23}$$

The spacecraft-planet direction $\hat{\mathbf{v}}$ expressed in **abc** is given by

$$\hat{\mathbf{v}} = -N\hat{\mathbf{r}} = \begin{bmatrix} v_a \\ v_b \\ v_c \end{bmatrix} \tag{24}$$

where v_a, v_b, and v_c are the direction cosines of $\hat{\mathbf{v}}$. The target-planet cone angle β is given by

$$\beta = \cos^{-1}(v_c) \tag{25}$$

and the target-planet clock angle α is given by

$$\alpha = \tan^{-1}(v_b/v_a) \tag{26}$$

V. Results of Application

The Near-Encounter Geometry Generation (NEGG) computer program, listed in the Appendix, was written to perform the previously defined computations. A 1969 Mars trajectory was simulated by NEGG and then compared to the SPACE output. Table 1 lists the trajectory elements and encounter geometry used as input to NEGG.

Table 2 lists the target-planet clock angle α and cone angle β, the spacecraft–target-planet range r, and the cone angle of Canopus β_c obtained from SPACE and NEGG for various times before encounter. Table 2 illustrates that NEGG can be used as a very effective mission analysis tool for the near-encounter phase of a trajectory. The shape and orientation of the trajectory can be changed by changing the input parameters B, θ, and V_∞ without significantly affecting the accuracy of the resultant trajectory geometry.

Table 1. Trajectory encounter characteristics

ξ_s, deg	ξ_c, deg	η_s, deg	η_c, deg	γ_p, deg	R_p, 10^6 km	B, 10^3 km	θ, deg	V_∞, km/s	C, 10^6 km^2/s
155.365	111.001	158.327	264.227	8.955	213.3	7.124	40.22	6.955	5469.7

Table 2. Geometry comparison

Days before encounter	SPACE program				NEGG program			
	α, deg	β, deg	r, 10^3 km	β_c, deg	α, deg	β, deg	r, 10^3 km	β_c, deg
20.0	127.985	165.110	121.800	77.914	127.949	164.900	120.900	77.915
10.0	118.339	160.340	61.020	77.566	118.305	160.240	60.490	77.566
5.0	115.539	157.873	30.030	77.445	115.535	157.849	30.270	77.445
1.0	114.931	156.155	6.091	77.372	114.930	156.154	6.081	77.372
0.0	231.607	94.817	6.318	77.357	231.607	94.817	6.318	77.357

VI. Recommendations

The largest error in the geometry generation resulted from assuming that the target planet had a constant rate about the sun. For the Mars trajectory, the angle λ described in Eq. (11) was in error by 0.05 deg at 10 days before encounter. Therefore, a more exact expression for $\dot{\lambda}$ would improve the geometry generation.

A perusal of the planet-centered conic section of the SPACE output can give insight into the validity of assuming fixed orbital parameters. Table 3 gives the hyperbolic orbital parameters B, θ, and V_∞ that describe the effective two-body trajectory solution for various times before Mars encounter for the selected trajectory.

Table 3. Effective two-body orbital parameters

Days before encounter	B, 10^3 km	θ, deg	V_∞, km/s
20.0	64.490	141.070	7.137
10.0	20.220	131.131	7.057
5.0	6.4371	62.955	7.011
1.0	7.1316	37.052	6.996
0.0	7.1426	36.876	6.995

Table 3 indicates that the assumption of fixed orbital parameters is not the best choice. An improvement in the geometry generation could be obtained by fitting second-order polynomials to the orbital parameters, such that

$$\left.\begin{aligned} B(\tau) &= B_0 + a_1\tau + a_2\tau^2 \\ \theta(\tau) &= \theta_0 + b_1\tau + b_2\tau^2 \\ V_\infty(\tau) &= V_\infty + c_1\tau + c_2\tau^2 \end{aligned}\right\} \qquad (27)$$

and then using the values of Eq. (27) with τ to solve Eq. (4), and then Eq. (1) for **r**. At more than a day from encounter, the direction to the planet (essentially along the incoming asymptote) is not affected strongly by the changing orbital parameters; therefore, the major improvement resulting from a polynomial representation of the orbital parameters would be in the spacecraft–target-planet range.

Nomenclature

a semimajor axis of hyperbola

$\mathbf{a}$ celestial vector to complete right-hand system **abc**

$\mathbf{b}$ celestial vector normal to **c** and **C**

B magnitude of **B**

$\mathbf{B}$ vector from center of target planet to incoming asymptote

$\mathbf{c}$ celestial vector from spacecraft to sun

C Kepler's area constant

$\mathbf{C}$ celestial vector from spacecraft to Canopus

e eccentricity of hyperbola

F analog of eccentric anomaly for hyperbola

i inclination of trajectory plane

$\mathbf{J}$ matrix which rotates sun direction about **W**

K transformation from **VUW** to **RST**

L transformation from **RST** to **XYZ**

M transformation from **VUW** to **XYZ**

N transformation from **RST** to **abc**

r magnitude of **r**

$\mathbf{r}$ target-planet–spacecraft vector in **RST**

$\mathbf{R}$ inertial vector to complete right-hand system **RST**

rad target-planet radius

R_p magnitude of $\mathbf{R}_p$

$\mathbf{R}_p$ target-planet–sun vector

$\mathbf{S}$ inertial vector along incoming asymptote

$\mathbf{T}$ inertial vector normal to **S** and parallel to the earth ecliptic plane

$\mathbf{U}$ inertial vector to complete right-hand system **VUW**

$\mathbf{v}$ spacecraft–target-planet vector in **abc**

$\mathbf{V}$ inertial target-planet–sun vector defined at encounter

v_a projection of **v** on **a**

v_b projection of **v** on **b**

v_c projection of **v** on **c**

$\mathbf{W}$ inertial vector directed above and normal to target-planet's plane of motion

$\mathbf{X}$ inertial vector equal to **V**

$\mathbf{Y}$ inertial vector equal to **S**

$\mathbf{Z}$ inertial vector normal to **X** and **Y**

α target-planet clock angle

β target-planet cone angle

β_c Canopus cone angle

γ_p angle between **S** and **V-U** plane

δ target-planet–spacecraft–Canopus angle

η true anomaly

η_c angle between $-\mathbf{T}$ and projection of **C** on **R-T** plane

η_s angle between $-\mathbf{T}$ and projection of $\mathbf{R}_p$ on **R-T** plane

θ direction of **B** in **R-T** plane

λ angle between $\mathbf{R}_p$ and **V**

$\dot{\lambda}$ time derivative of λ

μ target-planet gravitational constant

ξ_c angle between **S** and **C**

ξ_s angle between **S** and $\mathbf{R}_p$

σ_p angle between **V** and projection of **S** on **V-U** plane

τ time past encounter

ϕ target-planet phase angle

Φ target-planet angular diameter

ω argument of periapsis

ω' $\omega + \eta$

References

1. White, R. J., et al., *SPACE–Single Precision Cowell Trajectory Program,* Technical Memorandum 33-198. Jet Propulsion Laboratory, Pasadena, Calif., Jan. 15, 1965.

2. Ehricke, K. A., *Space Flight-Environmental and Celestial Mechanics.* D. Van Nostrand Co., Inc., New Jersey, Jan. 1960.

3. Traub, J. F., *Iterative Methods for the Solution of Equations.* Prentice-Hall, Inc., Englewood Cliffs, N. J., 1964.

4. Clarke, V. C., et al., *Design Parameters for Ballistic Interplanetary Trajectories, Part 1. One-Way Transfers to Mars and Venus,* Technical Report 32-77. Jet Propulsion Laboratory, Pasadena, Calif., Jan. 16, 1963.

Appendix

Near-Encounter Geometry Generation Program

```
      DIMENSION CN(3),TWR(3,2), RV(3),TIC(3,3),V(3),RPS(3)
      SN(X)=SIN(X*.17453293E-01)
      CS(X)=COS(X*.17453293E-01)
      U=.17453293E-01
      PI=3.1415927
1     READ(5,200) PMU, RAD, B, VH, C, RPP
      READ(5,201) ZAP, ZAC, ETS, ETC, CLP, GP, TH
      WW=C/(RPP*RPP)
      IF(C.LE.1.) WW=C
      AA=PMU/(VH*VH)
      EH=SQRT(1.+B*B/(AA*AA))
      PP=B*B/AA
      RP=PP/(1.+EH)
      HP=RP-RAD
      READ(5,201) TI, TF, DT
      NM=ABS((TI-TF)/DT)+1.95
      TI=TI*86400.
      TF=TF*86400.
      DT=DT*86400.
      WRITE(6,205) ZAP, ZAC, ETS, ETC, CLP, GP, TH
      WRITE(6,206) AA, B, HP, VH, C, RPP
      WRITE(6,202)
      WRITE(6,203)
      STH=SN(TH)
      CTH=CS(TH)
      SZP=SN(ZAP)
      CZP=CS(ZAP)
      SZC=SN(ZAC)
      CN(1)=-SZC*SN(ETC)
      CN(2)=CS(ZAC)
      CN(3)=-SZC*CS(ETC)
      SETS=SN(ETS)
      CETS=CS(ETS)
      SGP=SN(GP)
      CGP=CS(GP)
      CCLP=CZP/CGP
      SCLP=(RPP-2.E08)*SQRT(1.-CCLP*CCLP)/ABS(RPP-2.E08)
      IF(CLP.LT.0.) SCLP=-SCLP
      IF(CLP.GT.0.) SCLP=SN(CLP)
      TWR(1,1)=-SZP*SETS
      TWR(1,2)=(CZP*SETS*CGP*SCLP-CETS*SGP)/SZP
      TWR(2,1)=CZP
      TWR(2,2)=CGP*SCLP
      TWR(3,1)=-SZP*CETS
      TWR(3,2)=(CZP*CETS*CGP*SCLP+SETS*SGP)/SZP
      T=TI-DT
      DO 75 N=1,NM
      T=T+DT
      IF(T.GE.TF) T=TF
      IF(T.NE.0.) GO TO 17
      R=RP
      CET=1.
      SET=0.
      GO TO 20
17    R=RP+VH*ABS(T)
      CHF=(R+AA)/(AA*EH)
      SHF=SQRT(CHF*CHF-1.)
      F=ALOG(CHF+SHF)
18    FF=ABS(T)-AA*(EH*SHF-F)/VH
      FPF=-AA*(EH*CHF-1.)/VH
      F=F-FF/FPF
      Z=EXP(F)
      RT=AA*EH*(Z+1./Z)/2.-AA
      DR=ABS(RT-R)
      R=RT
      CHF=(R+AA)/(AA*EH)
      SHF=SQRT(CHF*CHF-1.)
      K=.43429448*ALOG(R)
      IF(10.**(K-5).LT.DR) GO TO 18
      CET=(PP/R-1.)/EH
      SET=SQRT(1.-CET*CET)*T/ABS(T)
```

```
20    SWP=(AA*CET+B*SET)/(AA*EH)
      CWP=(B*CET-AA*SET)/(AA*EH)
      RV(1)=R*CWP*STH
      RV(2)=R*SWP
      RV(3)=R*CWP*CTH
      DLT=ARCOS(-(RV(1)*CN(1)+RV(2)*CN(2)+RV(3)*CN(3))/R)/U
      PHI=2.*ARSIN(RAD/R)/U
C CALCULATION OF THE EFFECT OF THE TARGET PLANETS MOTION ABOUT THE SUN
      SWT=SIN(WW*T)
      CWT=COS(WW*T)
      DO 21 I=1,3
      RPS(I)=(TWR(I,1)*CWT+TWR(I,2)*SWT)*RPP
21    TIC(3,I)=RPS(I)-RV(I)
      PH=ARCOS((RPS(1)*RV(1)+RPS(2)*RV(2)+RPS(3)*RV(3))/(RPP*R))/U
      TICM=SQRT(TIC(3,1)*TIC(3,1)+TIC(3,2)*TIC(3,2)+TIC(3,3)*TIC(3,3))
      DO 23 I=1,3
23    TIC(3,I)=TIC(3,I)/TICM
      BC=ARCOS(TIC(3,1)*CN(1)+TIC(3,2)*CN(2)+TIC(3,3)*CN(3))/U
      TIC(2,1)=TIC(3,2)*CN(3)-TIC(3,3)*CN(2)
      TIC(2,2)=TIC(3,3)*CN(1)-TIC(3,1)*CN(3)
      TIC(2,3)=TIC(3,1)*CN(2)-TIC(3,2)*CN(1)
      TICM=SQRT(TIC(2,1)*TIC(2,1)+TIC(2,2)*TIC(2,2)+TIC(2,3)*TIC(2,3))
      DO 24 I=1,3
24    TIC(2,I)=TIC(2,I)/TICM
      TIC(1,1)=TIC(2,2)*TIC(3,3)-TIC(2,3)*TIC(3,2)
      TIC(1,2)=TIC(2,3)*TIC(3,1)-TIC(2,1)*TIC(3,3)
      TIC(1,3)=TIC(2,1)*TIC(3,2)-TIC(2,2)*TIC(3,1)
      DO 25 I=1,3
      V(I)=0.
      DO 25 J=1,3
25    V(I)=V(I)-TIC(I,J)*RV(J)/R
      AL=(.5*PI*(2.-V(2)/ABS(V(2)))-ATAN(V(1)/V(2)))/U
      BT=ARCOS(V(3))/U
      TDYS=T/86400.
      WRITE(6,204) TDYS, R, BT, AL, PH, PHI, DLT, BC
75    CONTINUE
      GO TO 1
200   FORMAT(4E10.1,2E12.4)
201   FORMAT(8E10.1)
202   FORMAT(1H0,2X,4HTIME,5X,5HRANGE,6X,4HCONE,5X,5HCLOCK,4X,5HPHASE,5X
     1,3HPHI,5X,5HDELTA,5X,2HBC)
203   FORMAT(1H ,1X,6H(DAYS),5X,4H(KM),6X,6(5H(DEG),4X))
204   FORMAT(1H ,F7.1,F12.1,6F9.3)
205   FORMAT(1H1,50X,22HTRAJECTORY INFORMATION/1X,4HZAP=,F8.3,3X,4HZAC=,
     1F8.3,3X,4HETS=,F8.3,3X,4HETC=,F8.3,3X,4HCLP=,F8.3,3X,3HGP=,F8.3,3X
     2,3HTH=,F8.3)
206   FORMAT(1H ,2HA=,F8.1,5X,2HB=,F8.1,5X,3HHP=,F8.1,5X,3HVH=,F7.3,5X,3
     1H C=,E11.4,5X,4HRPP=,E11.4)
      END
```